*Replication file 01a_Imposition_limited_sample_US
*Article: Counterfactual Coercion: Could harsher sanctions against Russia have prevented the worst?
*Authors: Thies Niemeier, Gerald Schneider


***************************************************************
***US***
***************************************************************

set seed 1234

*Prepare data
use "Dataset.dta", clear
keep if sender=="US"

** Filter for cases of importance
keep if pot_sanctioned_countries == 1

* Independent Variables
gen ln_oil_gas_value_2014 = ln(oil_gas_value_2014+1)
gen sender_colony=US_colony
gen sender_trade = ln_US_Trade_COW
gen coup_dummy = coup1
replace coup_dummy = 0 if coup_dummy == 1
replace coup_dummy = 1 if coup_dummy == 2

* Dependent variable: 1 if a threat or sanction case was ongoing in the dyad
gen sanction_threat = sanction_dyad
replace sanction_threat = 1 if threat_dyad==1
tab sanction_threat
gen sanction_train= sanction_threat if year < 2009
gen sanction_test= sanction_threat if year >= 2009

* lag time-series variables
sort ccodecow year
by ccodecow: gen l_v2x_polyarchy = v2x_polyarchy[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_gd_ptss = gd_ptss[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_coup_dummy = coup_dummy[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_one_sided_violence = one_sided_violence[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_conflict = conflict[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_mid_terr_integrity = mid_terr_integrity[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_ln_GDPpc_imputed = ln_GDPpc_imputed[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_sender_trade = sender_trade[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_ln_oil_gas_value = ln_oil_gas_value_2014[_n-1] if year==year[_n-1]+1
by ccodecow: gen l_defense_alliance = defense_alliance[_n-1] if year==year[_n-1]+1

* create dummy variables
tabulate l_gd_ptss, generate (pol_terr)

* sortieren nach Jahr, zur Vorbereitung RF model
gen u=0
replace u=1 if year >= 2009
sort u

** Imposition
* Random Forest Model
rforest sanction_threat l_v2x_polyarchy pol_terr* l_coup_dummy ///
l_one_sided_violence l_conflict l_mid_terr_integrity ///
l_ln_GDPpc_imputed l_sender_trade l_ln_oil_gas_value ///
sender_colony l_defense_alliance in 1/2789, type(class) iter(1500) numvars(15)

* Variable Importance
ereturn list
matrix list e(importance)
* write Variable importance to excel file
putexcel set "Supplemental_Material\Variable_Importance\Variable_Importance_Imposition_US_RF.xlsx", sheet("M") replace
putexcel A1=matrix(e(importance)), names

* Predictions
predict randonsUS
predict randonsUS0 randonsUS1, pr

* Confusion Matrix
* Sensitivity 53.2, Specificity 87.3
diagtest sanction_test randonsUS
tab2xl sanction_test randonsUS using Main_Article\US_Imposition_RF_Confusion_Matrix, row(1) col(1)

* Kappa .41
kap sanction_test randonsUS

* AUPR .52
prtab sanction_test randonsUS1, title("US", box bexpand) 
graph save "Graph" "Supplemental_Material\Prediction_Output\ROC-curves\AUPR_Imposition_RF_US.gph", replace


*******************************************************************
******* Compare to logistic regression ability ********************
*******************************************************************

sort ccodecow year
egen sum_sanction = sum(sanction_dyad), by(ccodecow)
gen dum_country = ccodecow
replace dum_country = 0 if sum_sanction==0
xtset ccodecow year

* Original
brglm sanction_train L.v2x_polyarchy i.L.gd_ptss i.L.coup_dummy i.L.one_sided_violence i.L.conflict i.L.mid_terr_integrity L.ln_GDPpc_imputed L.sender_trade L.ln_oil_gas_value_2014 i.L.sender_colony i.L.defense_alliance i.dum_country i.year, vce(cluster ccodecow)

*Predictions
predict onsUS_logistic
gen prob_sanc_onsUS_logistic = 1/(1+exp(-onsUS_logistic))
gen bin_prob_sanc_onsUS_logistic = cond(prob_sanc_onsUS_logistic > .5, 1,0)
replace bin_prob_sanc_onsUS_logistic =. if missing(prob_sanc_onsUS_logistic)

* Confusion Matrix
* Sensitivity 61.8, Specificity 86
diagtest sanction_test bin_prob_sanc_onsUS_logistic
tab2xl sanction_test bin_prob_sanc_onsUS_logistic using Supplemental_Material\Prediction_Output\Confusion_Matrixes\US_Imposition_PMLFE, col(1) row(1)

* Evaluation
*Kappa .41
kap sanction_test bin_prob_sanc_onsUS_logistic
* AUPR .64
prtab sanction_test prob_sanc_onsUS_logistic, title("US", box bexpand) 
graph save "Graph" "Supplemental_Material\Prediction_Output\ROC-curves\AUPR_Imposition_PMLFE_US.gph", replace


* Random forest only for cases with data
gen helper_sanction_test=sanction_test if!missing(prob_sanc_onsUS_logistic)
* Sensitivity 52.5, Specificity 86.7
diagtest helper_sanction_test randonsUS
tab2xl helper_sanction_test randonsUS using Supplemental_Material\Prediction_Output\Confusion_Matrixes\US_Imposition_RF_lim_sample, col(1) row(1)

* AUPR .51
prtab helper_sanction_test randonsUS1
graph save "Graph" "Supplemental_Material\Prediction_Output\ROC-curves\AUPR_Imposition_RF_US_lim_sample", replace

* Kappa .38
kap helper_sanction_test randonsUS
